

マイコンのための TSS 端末用プログラム

三 浦 定 俊

1. はじめに

最近、文化財資料においても情報検索のためのデータ・ベースづくりが話題になりつつある。情報検索とコンピュータは密接な関係を持っているが、今、多くの人を手軽に使えるようになったマイコン（マイクロ・コンピュータ）をもし文化財の情報整理に利用できたら大変便利であろうと考えられる。

マイクロ・コンピュータといっても、実際にはディスプレイ用のモニタ・テレビやプリンタ、外部記憶装置としてのカセット・テープまたはフロッピー・ディスクなどが組になっていて、ミニ・コンピュータよりも手軽な（といってもちょっと以前のミニコン以上の性能を持っているが）パーソナル・コンピュータ（パソコン）が主流である。例えば PET（コモドール）、APPLE II（アップル）、TRS-80（タンディ）、PC-8800（NEC）、FM-8（富士通）、MZ-80 B（シャープ）、M 223（ソード）、IF-800（沖電気）などがある。

マイコンは、個人的なデータ・ベースづくりや計算をやらせている内は良いが、博物館全体のデータ・ベース作成や画像処理などの大きな計算をやらせようとする、マイコンの記憶容量・計算速度の制限のために限界が生じる。このような場合には、マイコンを他所の大型計算機（ホスト・コンピュータ）と一般の電話回線をつなぎ、TSS 端末として使えば、いながらにして大型計算機を使用できる。ここではその一例として、筆者が現在使用しているフロッピー・ディスク付きの PET（CBM 3032）をインテリジェントな TSS 端末として使うためのプログラム（ベーシック言語）を公開して説明する。基本的なアイデアとして東大大型計算機センターの石田晴久氏のプログラムを参考にさせていただいた¹⁾²⁾。

なおこのプログラムは東大大型計算機センターとの交信のために作ったものなので、たとえ



図-1 使用中の TSS 端末概観

Fig. 1 Intelligent TSS terminal by CBM 3032

TSS 端末に PET を用いても、交信相手のホスト・コンピュータが違くと若干の手直しが必要である。

2. 電話回線との接続

マイコンを TSS 端末として使うためには、電話器との間にモデム（音響カプラー）を入れる必要がある、さらに音響カプラーとマイコンの接続に、PET では別に RS 232 C インターフェースを用意しなければならない³⁾。

TSS 端末の重要な機能に割り込みがあるが、PET にはそのための機能がないので、ここでは割り込みスイッチの付いた RS 232 C インターフェースを用いている。

3. プログラムの概説

このプログラムの大きさは、わかりやすくするために注釈文などを多く入れたので、ロードするのに約 6 キロバイト、走らせた時約 22 キロバイトとなり、小さなマイコンでは記憶容量がたりない。その時は行 140 に宣言しているメモリ・バッファのための配列 M% (8000) をもっと小さなものにすれば良い。配列の大きさ 8000 (8000 字分) で、大体 10 画面分の送受信コードの記憶ができる。

```

100 REM-----TSS PROGRAM "TERMINAL" FOR CBM3032-----
110 PRINT"INTELLI-TERMINAL PROGRAM FOR CBM 3032"
120 PRINT:PRINT"
130 MCX=0:REM-----MEMORY COUNTER
140 DIM M%(8000):REM-----BUFFER MEMORY
150 PRINT:PRINT"..BUFFER MEMORY OF 8000 BYTES RESERVED."
160 PRINT:PRINT"***** CONTROL KEY = SHIFT @ *****"
170 PRINT:PRINT" [ CODE LIST GIVEN BY 'H' ]"
180 PRINT:PRINT"..... ABLE TO RETURN BY 'GOTO 200'....."
190 REM-----MAIN LOOP-----
200 CLOSE1:CLOSE2:CLOSE3
210 OPEN 2,6,1:OPEN 3,6,6
220 PRINT:PRINT"....LOG-OFF MODE(RECEIVE & DISPLAY ONLY)"
230 REM-----GET A CHARACTER FROM KEYBOARD-----
240 GET S$:IF S#="" THEN 300
250 IF S#="-" THEN PRINT:GOTO 330
260 REM-----SEND A CHARACTER TO MODEM-----
270 PRINT S$:IF S#=CHR$(20) THEN PRINT#2,CHR$(0):GOTO 300
280 PRINT#2,S$;
290 REM-----RECEIVE A CHARACTER FROM MODEM-----
300 GET#3,R$:IF R#="" THEN 240
310 PRINT R$:GOTO 240
320 REM-----CONTROL CODES-----
330 CLOSE11:CLOSE12:PRINT:INPUT"CONTROL CODE-----":SS$
340 S=ASC(SS$)-63:IF S<1 OR S>21 THEN 380
350 IF S>10 THEN 370
360 ON S GOTO 520,380,430,380,450,430,430,490,420,380
370 ON S-10 GOTO 380,380,440,500,510,380,460,410,380,470,430
380 PRINT:PRINT"*** UNDEFINED CODE GIVEN INPUT AGAIN ***"
390 PRINT"----- CONTROL CODES' LIST GIVEN BY H":PRINT:GOTO 330
400 REM-----CALL SUBROUTINE-----
410 GOTO 220
420 GOSUB 530:GOTO 220
430 GOSUB 710:GOTO 220
440 GOSUB 770:GOTO 220
450 GOSUB 930:GOTO 220
460 GOSUB 1170:GOTO 220
470 GOSUB 1410:GOTO 220
480 GOSUB 1350:GOTO 220
490 GOSUB 1580:GOTO 220
500 GOSUB 1090:GOTO 220
510 MCX=0:PRINT:PRINT"..... MEMORY CLEARED .....":GOTO 220
520 GOSUB 1750:GOTO 220

```

図-2 プログラムリスト (1)

Fig. 2 Program list (1) [Main loop, command Q, N]

メモリ・バッファをM% (8000) と整数型配列にしたのはできるだけプログラムの大きさを小さくするためである。PET の中では、実数型配列で5バイト/字、文字型配列で3バイト/字、整数型配列で2バイト/字をそれぞれ必要とする。一番少なくともすむ整数型を選んだが、それでもこのメモリ・バッファ用配列のために16キロバイトものメモリが必要となった。

送受信の速さは、メモリ・バッファ用配列が整数型であるため、送受信した文字コードをメモリに格納する際、ASC 関数を使用して整数に変換しなければならない、300ボーがギリギリである。そのため TSS から受信する各行の先頭にある、復帰 (CR)、改行 (LF)、シフトイン (SI) のうち、必要のない SI をここでは除去できず、ファイルに書き込む時に除去するようにしてある (行1490)。

TSS で受信の際、PET の画面は1行40字であるので、1行80字で送信してくるホスト・コンピュータからのコードを、途中で改行する時、たまに1字分落してしまうことがある。これを防ぐために、行730で1行39字を指定するターミナル・コマンドをホスト・コンピュータに送れるようになっている。1行39字にしたのは、1行40字にすると40字までいった所で、PET 自身の改行とホスト・コンピュータ側からの改行を2度行うことになり、空行が1行できて見にくくなるためである。

4. 端末用コマンドの説明

このプログラムでは、TSS 端末用として14個のコマンドが定義され使えるようになっている。コマンドを使用するためには、コントロール・キーとしてここで用いている@のシフト (行250で"-"とあらわされている) を押せばいつでも割込みがかかってコマンドが使えるようになる。コマンドの説明はH (ヘルプ) で与えられる。もし定義された以外のコードを誤って押した場合は***UNDEFINED CODE GIVEN INPUT AGAIN***と表示され、もう一度入力が入らなければならない。

以下、プログラムにそって各コマンドの説明をする。

(1) コマンドH (行530—700)

画面がクリアされコマンドの種類が説明される。

(2) コマンドB (行720)

ホスト・コンピュータとの交信で最初に必要なログオン、ユーザ番号とパスワード (暗証) を送信する。各使用者自身の番号とパスワードを記入しなければならない。

(3) コマンドT (行730)

1行39字の指定を行う。

(4) コマンドE (行740)

交信を終了するためのログオフを送信する。

(5) コマンドL (行770—920)

TSS で送受信するすべてのコードをメモリに入れる (ログモード) コマンドである。このコマンドは TSS での交信記録を保存しておきたい時のためのもので、後に述べる表示のためのコマンドD、印刷のためのコマンドP、フロッピー・ディスクに書き込む (セーブ) ためのコマンドSなどすべて、このコマンドを用いた後に使える。もしメモリ・バッファが空の状態でもD・P・Sなどのコマンドを使おうとすると MEMORY EMPTY の表示がでて画面に表示するだけ (ログオフ・モード) になる。

メモリ・バッファが一杯になった時も、MEMORY FULL が画面に表示され、ログオフ

```

530 REM-----SUBROUTINE (1) HELP-----
540 REM-----MENU-----
550 PRINT:PRINT"THE FOLLOWING CONTROL CODES CAN BE USED"
560 PRINT:PRINT" B=BEGIN(SEND LOGON,USER ID/PASS WORD)"
570 PRINT" D=DISPLAY MEMORY"
580 PRINT" E=END(SEND LOGOFF)"
590 PRINT" F=FILE DEFINITION"
600 PRINT" G=GET FILE & SEND TO MODEM"
610 PRINT" H=HELP,LIST MENU"
620 PRINT" L=LOG-MODE(RECEIVE & MEMORY)"
630 PRINT" M=MEMORY CHECK"
640 PRINT" N=NEW,CLEAR MEMORY"
650 PRINT" P=PRINT MEMORY"
660 PRINT" Q=QUIT LOG-MODE"
670 PRINT" S=SAVE MEMORY INTO FILE"
680 PRINT" T=TERMINAL COMMAND(LINE-SIZE=39)"
690 PRINT" @=FILE ERROR MESSAGE"
700 RETURN
710 REM-----SUBROUTINE (2) COMMAND -----
720 IF SS#="B" THEN PRINT".....BEGIN LOGON":B#="LOGON ?????/?????????":GOTO 750
730 IF SS#="T" THEN PRINT".....LINESIZE=39":B#="TERM LINES(39)":GOTO 750
740 PRINT".....END LOGOFF":B#="LOGOFF"
750 LC#=LEN(B#):FOR I=1 TO LC#:S#="MID$(B#,I,1):PRINT#2,S#:NEXT I
760 PRINT#2:RETURN
770 REM-----SUBROUTINE (3) LOG-MODE-----
780 IF MC#=0 THEN PRINT:PRINT"..... MEMORY EMPTY .....":PRINT:GOTO 810
790 GOSUB 1090:REM-----MEMORY CHECK
800 REM-----RECEIVE & SEND-----
810 KX=MC#
820 FOR I=KX TO 8000
830 GET#3,R#:IF R#="" THEN 850
840 PRINT R#:MX(MC#)=ASC(R#):GOTO 900
850 GET S#:IF S#="" THEN 830
860 IF S#="-" THEN 320
870 PRINT S#:MX(MC#)=ASC(S#)
880 IF S#=CHR$(20) THEN PRINT#2,CHR$(8):GOTO 900
890 PRINT#2,S#
900 MC#=MC#+1:NEXT I
910 REM-----MEMORY FULL-----
920 PRINT:PRINT:PRINT"..... MEMORY FULL .....":RETURN

```

図-3 プログラムリスト (2)

Fig. 3 Program list (2) [command H, B, T, E, L]

・モードに戻るが、この時ホスト・コンピュータから受信中であると、5文字分ほどのコードを落してしまう。それを防ぐには、後に述べるメモリチェックのためのコマンドMを使って、時々メモリ・バッファの空きを確かめればよい。

(6) コマンドD (行930—1050)

メモリの内容を画面に表示する。途中でとめたい時にはコントロール・キーを押す。

行1060—1080はコントロール・キーが押されていないかチェックするためのサブルーチンで他のコマンド実行中にもこのサブルーチンはよびだされて、コントロール・キーによる中断がいつでもできるようになっている。

メモリ内容の表示が終了した時は、その内容をフロッピー・ディスクに書き込むか、消去するか、保持するか聞いてくる。

(7) コマンドQ (行410)

ログ・モードからログオフ・モードにもどりたい時や、メモリの内容を表示、印刷、書き込みしている途中にコントロール・キーで強制的に中断させ、ログオフ・モードにもどりたい時、このコマンドを使う。ただし、プログラムを走らせた最初やログ・モード以外のコマンドでの実行が終了した時は、いつもログオフ・モードにもどるようになっている。

(8) コマンドM (行1090—1160)

```

930 REM-----SUBROUTINE (4) DISPLAY MEMORY-----
940 IF MCX=0 THEN PRINT:PRINT"..... MEMORY EMPTY .....":RETURN
950 IF MCX>8000 THEN MCX=8000
960 PRINT:PRINT"...";MCX;"CHARACTERS WILL BE DISPLAYED."
970 FOR I=0 TO MCX
980 IF M%(I)=15 THEN 1010
990 PRINT CHR$(M%(I));
1000 GOSUB 1060:REM-----CHECK KB
1010 NEXT I
1020 IF MCX>=8000 THEN PRINT:PRINT"..... MEMORY FULL .....":GOTO 1040
1030 PRINT:PRINT"...";8000-MCX;"BYTES STILL FREE....."
1040 GOSUB 1090:REM-----MEMORY CHECK
1050 RETURN
1060 REM-----SUBROUTINE (5) CHECK KEYBOARD-----
1070 GET S$:IF S#="-" THEN 320
1080 RETURN
1090 REM-----SUBROUTINE (6) MEMORY CHECK-----
1100 IF MCX=0 THEN PRINT:PRINT"..... MEMORY EMPTY .....":RETURN
1110 PRINT:PRINT"...";MCX;"CHARS IN MEMORY...."
1120 PRINT:INPUT"SAVE,CLEAR OR KEEP (0,1 OR 2)";NX
1130 IF NX=0 THEN GOSUB 1410:GOTO 1160:REM-----SAVE MEMORY
1140 IF NX=1 THEN MCX=0
1150 IF MCX>=8000 THEN PRINT:PRINT"..... MEMORY FULL .....":RETURN
1160 PRINT:PRINT"...";8000-MCX;"BYTES FREE.....":RETURN
1170 REM-----SUBROUTINE (7) PRINT MEMORY-----
1180 IF MCX=0 THEN PRINT:PRINT"..... MEMORY EMPTY .....":RETURN
1190 IF MCX>8000 THEN MCX=8000
1200 PRINT:PRINT"...";MCX;"CHARACTERS WILL BE PRINTED.. "
1210 CLOSE 4:OPEN 4,4
1220 FOR I=0 TO MCX
1230 IF M%(I)=10 OR M%(I)=15 THEN 1270
1240 IF M%(I)=20 THEN PRINT#4,"*";:GOTO 1270
1250 PRINT#4,CHR$(M%(I));
1260 GOSUB 1060:REM-----CHECK KB
1270 NEXT I
1280 PRINT#4:CLOSE 4
1290 PRINT:PRINT"..... END OF DATA ....."
1300 IF MCX>=8000 THEN PRINT:PRINT".....MEMORY FULL.....":GOTO 1330
1310 REM-----MEMORY CHECK-----
1320 PRINT:PRINT"...";8000-MCX;"BYTES STILL FREE.":PRINT
1330 GOSUB 1090:REM-----MEMORY CHECK
1340 RETURN

```

図-4 プログラムリスト (3)

Fig. 4 Program list (3) [command D, M, P]

メモリ・バッファ内の空きを調べる。もし中身があれば、それをフロッピー・ディスクに書き込むか、消去するか、保持するか聞いてくる。保持すると以後の送受信記録はそれまでのものにつながってメモリ・バッファ内に保存されて行く。

(9) コマンド N (行510)

単に、メモリを消去したい時に用いる。

(10) コマンド P (行1170—1340)

メモリの内容をプリンタに印刷させる。プリンタにバック・スペースの機能がないので、キーボードで消去キーが押された時は、←を印刷することにしてある(行1240)。また東大大型センターからのデータには、各行に必ず CR と LF がついていて、そのままにしておく一行おきの印刷になってしまうので、行1230で LF は SI と共に除いている。

(11) コマンド F (行1350—1400)

フロッピー・ディスクのファイルを定義する。書き込み用ファイルのディスクはドライブナンバー 0、読み込み用は 1 と決められている。以前に用いたファイルでも再書き込みができる。もしこのコマンドでファイルをあらかじめ定義せずに、ファイルにメモリを書き込もうとしたり、ファイルからデータを読み取ろうとすると、FILE NAME IS NOT DEFINED と表示されてログオフ・モードになるので、その時はあらためてコントロール・キーを押した上

```

1350 REM-----SUBROUTINE (8) FILE DEFINITION-----
1360 PRINT:INPUT"RECEIVE-FILE NAME(DR=0)=";RF#
1370 INPUT"SEND-FILE NAME(DR=1)=";SF#
1380 RF#="@0:"+RF#;SF#="1:"+SF#
1390 CLOSE1:OPEN 1,8,15:PRINT#1,"I"
1400 PRINT:PRINT"FILES [";RF#;" AND ";SF#;"] ARE DEFINED":RETURN
1410 REM-----SUBROUTINE (9) SAVE MEMORY-----
1420 IF RF#="" THEN PRINT:PRINT"..... FILE NAME IS NOT DEFINED .....":RETURN
1430 IF MC%=0 THEN PRINT:PRINT"..... MEMORY EMPTY .....":RETURN
1440 IF MC%>8000 THEN MC%=8000
1450 REM-----SAVE MEMORY TO RECEIVE-FILE-----
1460 CLOSE11:OPEN 11,8,2,RF#+",.SEQ,WRITE"
1470 PRINT:PRINT"...";MC%:"CHARS WILL BE SAVED IN FILE"
1480 FOR I=0 TO MC%
1490 IF M%(I)=15 THEN 1520
1500 PRINT#11,CHR%(M%(I));
1510 GOSUB 1060:REM-----CHECK KB
1520 NEXT I
1530 CLOSE11
1540 PRINT:PRINT"...";MC%:"CHARS SAVED IN FILE ";RF#
1550 PRINT:INPUT"MEMORY CLEAR OR KEEP(1 OR 2)";M%
1560 IF M%=2 THEN RETURN
1570 MC%=0:PRINT:PRINT"..... MEMORY CLEARED .....":RETURN
1580 REM-----SUBROUTINE (10) GET FILE-----
1590 IF SF#="" THEN PRINT:PRINT"..... FILE NAME IS NOT DEFINED .....":RETURN
1600 CLOSE12:OPEN 12,8,3,SF#+",.SEQ,READ"
1610 REM-----SEND FILE TO MODEM-----
1620 PRINT:PRINT"...DATA WILL BE SENT TO MODEM FROM FILE ";SF#
1630 PR#=""
1635 PRINT:PRINT"ENTER PROMPT-CODE[0(BULK),END,(SKIP) OR OTHER] AT EVERY CR"
1640 IF PR#="0" THEN 1660
1650 INPUT PR#:IF PR#="END" THEN 1740
1660 GET#12,FS#:IF ST>0 THEN 1740
1663 PRINT FS#;
1665 IF PR#<>" THEN 1670
1667 IF FS#=CHR%(13) THEN PRINT".....THIS LINE WAS SKIPPED...":GOTO 1650
1669 GOTO 1660
1670 PRINT#2,FS#;
1680 GOSUB 1060:REM-----CHECK KB
1690 IF FS#=CHR%(13) THEN T=TI:GOTO 1710
1700 GOTO 1660
1710 GET#3,R#:PRINT R#:IF R#<>" THEN T=TI:GOTO 1710
1720 IF TI-T<60 THEN 1710
1730 GOTO 1640
1740 CLOSE12:PRINT:PRINT"..... END OF DATA .....":RETURN
1750 REM-----SUBROUTINE (11) FILE ERROR MESSAGE-----
1760 CLOSE1:OPEN 1,8,15
1770 INPUT#1,A#,B#,C#,D#
1780 PRINT:PRINT A#;" ";B#,C#,D#:RETURN

```

図-4 プログラムリスト (4)

Fig. 4 Program list (4) [command F, S, G, @]

で、このコマンドFでファイルの定義をすればよい。

(12) コマンドS (行1410—1570)

メモリの内容をファイルに書き込む。途中コントロール・キーを押すと中断され、ログオフ・モードにもどるが、その場合でも画面に表示されたメモリの内容まではファイルに書き込まれている。もう一度コマンドSを使うと、またメモリの最初から書き込まれていく。

書き込みが終了するとメモリの内容を消去するか保持するか聞いてくるが、この後印刷させるつもりがないなら消去しておいた方が、メモリ・バッファをあふれさせないために賢明である。

(13) コマンドG (行1580—1740)

ファイルの内容をモデム (ホスト・コンピュータ) へ送信する。最初にプロンプト・コードをきいてくるが、0と答えるとファイルの内容を最後までどんどん送信 (バルク出力), ENDと答えると送信終了; /なら、その行は送信せず読みとばし、その他の文字 (何でも良い) を

入れると一行ずつ送信される。途中からバルク出力にすることも可能である。送信の方法は 1 行送信した後、1 秒待ってもしホスト・コンピュータからのメッセージがなければ、また送信を開始（バルク出力の時は自動的に、その他ならまたプロンプト・コードを受けてから）する。ホスト・コンピュータからのメッセージがある時は、それを受けた後、1 秒たってまた送信を開始する。ここで受けるメッセージはたいいていエラー・メッセージであって短いものである。

(14) コマンド@

フロッピー・ディスクを使っていてエラーの表示がでた時に、フロッピーからのエラー・メッセージを見るためのコマンドである。

この他、プログラムを中断して PET のベーシックに戻るためにブレイク・キーも使える。再開する時に GOTO 200 とすれば、メモリの内容を失わずにそのまま TSS モードに戻ることができる。この TSS プログラムを走らせる前にあらかじめ PET のフロッピー・ディスク 3040 についている DOS サポートを入れておき、省略されたコマンドを使えるようにしておく、TSS で交信中にフロッピー・ディスクのディレクトリを調べてファイルを確認することができたりして、大変便利である。

このプログラムはベーシックで書かれているために大きさや実行速度など苦しい点が多い。もしアセンブラでプログラムできたらずっと楽になると考えられる。

参 考 文 献

- 1) 石田晴久：マイコンをインテリ端末にするための BASIC プログラム，東大大型計算機センターニュース，12，8 (1980)，pp. 46—53
- 2) 石田晴久：インテリ端末用 BASIC プログラムの使い方，東大大型計算機センターニュース，12，9 (1980)，pp. 18—21
- 3) ACTAM 330-MIII 音響カプラ概要説明書（田村電機製作所）

Basic Program for an Intelligent TSS Terminal by Micro-Computer

Sadatoshi MIURA

A program for a PET (CBM 3032) with a floppy disk to be an intelligent TSS terminal is listed. In the program 15 commands can be used with a control key (shift of @). "H" gives an explanation of the commands. "B", "T" and "E" send special commands to a host-computer. "L" stores all TSS codes to a buffer memory (8000 bytes) of PET (logon-mode). By "Q" command you can quit the logon-mode to a logoff-mode (only display). "M" checks the buffer memory and "N" clears it. You can print out the content of the buffer memory by "P". Commands for the floppy disk are "F" (file definition), "S" (save the content of the buffer memory to the disk), "G" (get a file from the disk and send it to the host-computer) and "@" (ask an error message of the floppy disk).

Some changes of this program may be necessary according to each TSS system of a different host-computer.